Pensieve Header: β-Calculus rev. 2; rev. 3 in pensieve://2012-03/.

Utilities

```
βSimplify = Factor;
SetAttributes[βCollect, Listable];
βCollect[B[ω_, μ_]] := B[
   βSimplify[ω],
   Collect[μ, _h, Collect[#, _t, βSimplify] &]
  ];
(* "L" for "Labels" *)
hL[β_] := Union[Cases[β, h[s_] :→ s, Infinity]];
tL[β_] := Union[Cases[β, t[s_] | c_s_ :→ s, Infinity]];
dL[β_] := Union[hL[β], tL[β]];
βForm[B[ω_, μ_]] := Module[
   {tails, heads, mat},
   tails = tL[B[ω, μ]]; heads = hL[B[ω, μ]];
   mat = Outer[βSimplify[Coefficient[μ, h[#1] t[#2]]] &, heads, tails];
   PrependTo[mat, t /@ tails];
   mat = Prepend[Transpose[mat], Prepend[h /@ heads, ω]];
   MatrixForm[mat]
  ];
βForm[else_] := else /. β_B :→ βForm[β];
B /: B[ω1_, β1_] == B[ω2_, β2_] := (ω1 == ω2) && (β1 == β2);
```

The Meta-Cross-Product

The “Tails” meta-group

```
tm[x_, y_, z_][β_] := β /. {t[x] → t[z], t[y] → t[z], c_x → c_z, c_y → c_z};
tΔ[z_, x_, y_][β_] := βCollect[β /. {t[z] → t[x] + t[y], c_z → c_x + c_y}];
tη[x_][β_] := βCollect[(β /. t[x] → 0) /. c_x → 0];
tS[x_][β_] := βCollect[β /. {t[x] → -t[x], c_x → -c_x}];
tA[_][β_] := βCollect[β];
tP[rules___Rule][β_] := βCollect[
   β /. {t[x_] :→ t[x /. {rules}], c_x_ :→ c_(x /. {rules})}
  ];
```

The “Heads” meta-group

```
hm[x_, y_, z_][B[ω_, μ_]] := Module[
   {γx = D[μ, h[x]], γy = D[μ, h[y]], M = μ /. h[x] | h[y] → 0},
   B[ω, M + h[z] (γx + γy + (γx /. t[i_] :→ c_i) γy)] // βCollect
  ];
hΔ[z_, x_, y_][β_] := βCollect[β /. h[z] → h[x] + h[y]];
hη[x_][β_] := βCollect[β /. h[x] → 0];
hS[x_][B[ω_, μ_]] := Module[{γ},
   γ = 1 + D[μ, h[x]] /. t[s_] :→ c_s;
   βCollect[B[ω, μ /. h[x] → -h[x] / γ]]
  ];
hA[x_][β_] := hS[x][β];
hP[rules___Rule][β_] := βCollect[β /. h[x_] :→ h[x /. {rules}]];
```

The TH → HT Swap

```
thswap[x_, y_][B[ω_, μ_]] := Module[
   {α, β, γ, δ, ϵ},
   α = Coefficient[μ, h[y] t[x]];
   β = D[μ, t[x]] /. h[y] → 0;
   γ = D[μ, h[y]] /. t[x] → 0;
   δ = μ /. h[y] | t[x] → 0;
   ϵ = 1 + c_x α;
   B[ω * ϵ, Plus[
      α (1 + (γ /. t[i_] :→ c_i) / ϵ) h[y] t[x],
      β (1 + (γ /. t[i_] :→ c_i) / ϵ) t[x],
      γ / ϵ h[y],
      δ - c_x / ϵ γ * β
     ]] // βCollect
  ];
```

The “double” meta-group

```
dm[x_, y_, z_][β_] := β // thswap[x, y] // hm[x, y, z] // tm[x, y, z];
dΔ[z_, x_, y_][β_] := β // tΔ[z, x, y] // hΔ[z, x, y];
```

The “external” product

```
B /: B[ω1_, μ1_] B[ω2_, μ2_] := B[ω1 * ω2, μ1 + μ2];
```

“Braid-Like” operations

```
Unprotect[NonCommutativeMultiply];
β_ ** ν_ := Module[
   {ρ, σ, labels},
   ρ = β * (ν /. {h[s_] :> h[σ[s]], t[s_] :> t[σ[s]], c_s_ :> c_σ[s]});
   labels = Union[Cases[{β, ν}, h[s_] | t[s_] | c_s_ :> s, Infinity]];
   Do[
    ρ = ρ // dm[s, σ[s], s],
    {s, labels}
   ];
   ρ
  ];
B /: Inverse[B[ω_, μ_]] := Module[
   {ρ = B[1, μ]},
   Do[ρ = ρ // dA[s], {s, Union[hL[ρ], tL[ρ]]}];
   ReplacePart[ρ, 1 → 1 / ω] // βCollect
  ];
```

The R-Matrix

```
R[x_, y_] := B[1, (E^c_x - 1) / c_x * t[x] h[y]];
Ri[x_, y_] := B[1, (E^(-c_x) - 1) / c_x * t[x] h[y]];
```

Testing the meta-cross-product axioms

The “T” meta-group

```
{
   β = B[ω[c1, c2, c3, c4], Sum[αi[c1, c2, c3, c4] t[i] h[1], {i, 4}]],
   β // tm[1, 2, 1],
   t1 = β // tm[1, 2, 1] // tm[1, 3, 1],
   t2 = β // tm[2, 3, 28] // tm[1, 28, 1],
   t1 == t2
  } // βForm // ColumnForm
```

$\omega[c_1, c_2, c_3, c_4]$	$h[1]$
$t[1]$	$\alpha_1[c_1, c_2, c_3, c_4]$
$t[2]$	$\alpha_2[c_1, c_2, c_3, c_4]$
$t[3]$	$\alpha_3[c_1, c_2, c_3, c_4]$
$t[4]$	$\alpha_4[c_1, c_2, c_3, c_4]$

$\omega[c_1, c_1, c_3, c_4]$	$h[1]$
$t[1]$	$\alpha_1[c_1, c_1, c_3, c_4] + \alpha_2[c_1, c_1, c_3, c_4]$
$t[3]$	$\alpha_3[c_1, c_1, c_3, c_4]$
$t[4]$	$\alpha_4[c_1, c_1, c_3, c_4]$

$\omega[c_1, c_1, c_1, c_4]$	$h[1]$
$t[1]$	$\alpha_1[c_1, c_1, c_1, c_4] + \alpha_2[c_1, c_1, c_1, c_4] + \alpha_3[c_1, c_1, c_1, c_4]$
$t[4]$	$\alpha_4[c_1, c_1, c_1, c_4]$

$\omega[c_1, c_1, c_1, c_4]$	$h[1]$
$t[1]$	$\alpha_1[c_1, c_1, c_1, c_4] + \alpha_2[c_1, c_1, c_1, c_4] + \alpha_3[c_1, c_1, c_1, c_4]$
$t[4]$	$\alpha_4[c_1, c_1, c_1, c_4]$

True

The “H” meta-group

```
{
  β = B[ω, Sum[α_{10 i+j} t[i] h[j], {i, 2}, {j, 4}]],
  β // hm[1, 2, 1],
  t1 = β // hm[1, 2, 1] // hm[1, 3, 1],
  t2 = β // hm[2, 3, 28] // hm[1, 28, 1],
  t1 == t2
 } // βForm // ColumnForm
```

ω	h[1]	h[2]	h[3]	h[4]
t[1]	α_{11}	α_{12}	α_{13}	α_{14}
t[2]	α_{21}	α_{22}	α_{23}	α_{24}

ω	h[1]	h[3]	h[4]
t[1]	$\alpha_{11} + \alpha_{12} + C_1 \alpha_{11} \alpha_{12} + C_2 \alpha_{12} \alpha_{21}$	α_{13}	α_{14}
t[2]	$\alpha_{21} + \alpha_{22} + C_1 \alpha_{11} \alpha_{22} + C_2 \alpha_{21} \alpha_{22}$	α_{23}	α_{24}

ω	h[1]
t[1]	$\alpha_{11} + \alpha_{12} + C_1 \alpha_{11} \alpha_{12} + \alpha_{13} + C_1 \alpha_{11} \alpha_{13} + C_1 \alpha_{12} \alpha_{13} + C_1^2 \alpha_{11} \alpha_{12} \alpha_{13} + C_2 \alpha_{12} \alpha_{21} + C_2 \alpha_{13} \alpha_{21} + C_1 C_2 \alpha_{12}$
t[2]	$\alpha_{21} + \alpha_{22} + C_1 \alpha_{11} \alpha_{22} + C_2 \alpha_{21} \alpha_{22} + \alpha_{23} + C_1 \alpha_{11} \alpha_{23} + C_1 \alpha_{12} \alpha_{23} + C_1^2 \alpha_{11} \alpha_{12} \alpha_{23} + C_2 \alpha_{21} \alpha_{23} + C_1 C_2 \alpha_{12}$

ω	h[1]
t[1]	$\alpha_{11} + \alpha_{12} + C_1 \alpha_{11} \alpha_{12} + \alpha_{13} + C_1 \alpha_{11} \alpha_{13} + C_1 \alpha_{12} \alpha_{13} + C_1^2 \alpha_{11} \alpha_{12} \alpha_{13} + C_2 \alpha_{12} \alpha_{21} + C_2 \alpha_{13} \alpha_{21} + C_1 C_2 \alpha_{12}$
t[2]	$\alpha_{21} + \alpha_{22} + C_1 \alpha_{11} \alpha_{22} + C_2 \alpha_{21} \alpha_{22} + \alpha_{23} + C_1 \alpha_{11} \alpha_{23} + C_1 \alpha_{12} \alpha_{23} + C_1^2 \alpha_{11} \alpha_{12} \alpha_{23} + C_2 \alpha_{21} \alpha_{23} + C_1 C_2 \alpha_{12}$

```
True
```

```
{
  β = B[ω, Sum[α10 i+j[c1, c2] * t[i] h[j], {i, 2}, {j, 2}]],
  β // tΔ[2, 2, 3],
  β // hΔ[2, 2, 3],
  β // hΔ[2, 2, 3] // hS[3],
  β // hΔ[2, 2, 3] // hS[3] // hm[2, 3, 2],
  β // hΔ[2, 2, 3] // hS[3] // hm[3, 2, 2],
  β  // hS[1],
  β  // hS[1] // hS[1]
 } // βForm // ColumnForm
```

$$\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & \alpha_{11}[c_1, c_2] & \alpha_{12}[c_1, c_2] \\ t[2] & \alpha_{21}[c_1, c_2] & \alpha_{22}[c_1, c_2] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & \alpha_{11}[c_1, c_2 + c_3] & \alpha_{12}[c_1, c_2 + c_3] \\ t[2] & \alpha_{21}[c_1, c_2 + c_3] & \alpha_{22}[c_1, c_2 + c_3] \\ t[3] & \alpha_{21}[c_1, c_2 + c_3] & \alpha_{22}[c_1, c_2 + c_3] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11}[c_1, c_2] & \alpha_{12}[c_1, c_2] & \alpha_{12}[c_1, c_2] \\ t[2] & \alpha_{21}[c_1, c_2] & \alpha_{22}[c_1, c_2] & \alpha_{22}[c_1, c_2] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11}[c_1, c_2] & \alpha_{12}[c_1, c_2] & -\frac{\alpha_{12}[c_1,c_2]}{1+c_1\,\alpha_{12}[c_1,c_2]+c_2\,\alpha_{22}[c_1,c_2]} \\ t[2] & \alpha_{21}[c_1, c_2] & \alpha_{22}[c_1, c_2] & -\frac{\alpha_{22}[c_1,c_2]}{1+c_1\,\alpha_{12}[c_1,c_2]+c_2\,\alpha_{22}[c_1,c_2]} \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] \\ t[1] & \alpha_{11}[c_1, c_2] \\ t[2] & \alpha_{21}[c_1, c_2] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] \\ t[1] & \alpha_{11}[c_1, c_2] \\ t[2] & \alpha_{21}[c_1, c_2] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & -\frac{\alpha_{11}[c_1,c_2]}{1+c_1\,\alpha_{11}[c_1,c_2]+c_2\,\alpha_{21}[c_1,c_2]} & \alpha_{12}[c_1, c_2] \\ t[2] & -\frac{\alpha_{21}[c_1,c_2]}{1+c_1\,\alpha_{11}[c_1,c_2]+c_2\,\alpha_{21}[c_1,c_2]} & \alpha_{22}[c_1, c_2] \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & \alpha_{11}[c_1, c_2] & \alpha_{12}[c_1, c_2] \\ t[2] & \alpha_{21}[c_1, c_2] & \alpha_{22}[c_1, c_2] \end{pmatrix}$$

```
{
  β = B[ω, Sum[α_{10 i+j} * t[i] h[j], {i, 2}, {j, 3}]],
  t1 = β // hm[1, 2, 1] // hS[1],
  t2 = β // hS[1] // hS[2] // hm[2, 1, 1],
  t1 == t2 // Simplify
 } // βForm // ColumnForm
```

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11} & \alpha_{12} & \alpha_{13} \\ t[2] & \alpha_{21} & \alpha_{22} & \alpha_{23} \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[3] \\ t[1] & -\frac{\alpha_{11}+\alpha_{12}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21}}{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})} & \alpha_{13} \\ t[2] & -\frac{\alpha_{21}+\alpha_{22}+c_1\,\alpha_{11}\,\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22}}{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})} & \alpha_{23} \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[3] \\ t[1] & -\frac{\alpha_{11}+\alpha_{12}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21}}{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})} & \alpha_{13} \\ t[2] & -\frac{\alpha_{21}+\alpha_{22}+c_1\,\alpha_{11}\,\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22}}{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})} & \alpha_{23} \end{pmatrix}$$

True

Testing “thswap”

```
Clear[β];
{β1 = B[ω, h[1] t[1] α + h[2] t[1] β + h[1] t[2] γ + h[2] t[2] δ],
  β1 // thswap[1, 1]
 } // βForm
```

$$\left\{\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & \alpha & \beta \\ t[2] & \gamma & \delta \end{pmatrix}, \begin{pmatrix} \omega\,(1+\alpha\,c_1) & h[1] & h[2] \\ t[1] & \frac{\alpha\,(1+\alpha\,c_1+\gamma\,c_2)}{1+\alpha\,c_1} & \frac{\beta\,(1+\alpha\,c_1+\gamma\,c_2)}{1+\alpha\,c_1} \\ t[2] & \frac{\gamma}{1+\alpha\,c_1} & \frac{\delta-\beta\,\gamma\,c_1+\alpha\,\delta\,c_1}{1+\alpha\,c_1} \end{pmatrix}\right\}$$

```
{
  β = B[ω, Sum[α_{10 i+j} t[i] h[j], {i, 2}, {j, 3}]],
  β // hm[1, 2, 1],
  t1 = β // hm[1, 2, 1] // thswap[1, 1],
  t2 = β // thswap[1, 1] // thswap[1, 2] // hm[1, 2, 1],
  t1 == t2 // Simplify
 } // βForm // ColumnForm
```

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] \\ t[1] & \alpha_{11} & \alpha_{12} & \alpha_{13} \\ t[2] & \alpha_{21} & \alpha_{22} & \alpha_{23} \end{pmatrix}$$

$$\begin{pmatrix} \omega & h[1] & h[3] \\ t[1] & \alpha_{11}+\alpha_{12}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21} & \alpha_{13} \\ t[2] & \alpha_{21}+\alpha_{22}+c_1\,\alpha_{11}\,\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22} & \alpha_{23} \end{pmatrix}$$

$$\begin{pmatrix} \omega\,\left(1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}\right) & h[1] & \\ t[1] & \frac{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(\alpha_{11}+\alpha_{12}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}{1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}} & \\ t[2] & \frac{\alpha_{21}+\alpha_{22}+c_1\,\alpha_{11}\,\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22}}{1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}} & -c_1\,\alpha_1 \end{pmatrix}$$

$$\begin{pmatrix} \omega\,\left(1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}\right) & h[1] & \\ t[1] & \frac{(1+c_1\,\alpha_{11}+c_2\,\alpha_{21})\,(\alpha_{11}+\alpha_{12}+c_1\,\alpha_{11}\,\alpha_{12}+c_2\,\alpha_{12}\,\alpha_{21})\,(1+c_1\,\alpha_{12}+c_2\,\alpha_{22})}{1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}} & \\ t[2] & \frac{\alpha_{21}+\alpha_{22}+c_1\,\alpha_{11}\,\alpha_{22}+c_2\,\alpha_{21}\,\alpha_{22}}{1+c_1\,\alpha_{11}+c_1\,\alpha_{12}+c_1^2\,\alpha_{11}\,\alpha_{12}+c_1\,c_2\,\alpha_{12}\,\alpha_{21}} & -c_1\,\alpha_1 \end{pmatrix}$$

True

```
{
  β = B[ω, Sum[α10 i+j t[i] h[j], {i, 3}, {j, 2}]],
  t1 = β // tm[1, 2, 1] // thswap[1, 1],
  t2 = β // thswap[2, 1] // thswap[1, 1] // tm[1, 2, 1],
  t1 == t2 // Simplify
 } // βForm // ColumnForm
```

$$\begin{pmatrix} \omega & h[1] & h[2] \\ t[1] & \alpha_{11} & \alpha_{12} \\ t[2] & \alpha_{21} & \alpha_{22} \\ t[3] & \alpha_{31} & \alpha_{32} \end{pmatrix}$$

$$\begin{pmatrix} \omega\,(1 + C_1\,\alpha_{11} + C_1\,\alpha_{21}) & h[1] & h[2] \\ t[1] & \frac{(\alpha_{11}+\alpha_{21})\,(1+C_1\,\alpha_{11}+C_1\,\alpha_{21}+C_3\,\alpha_{31})}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} & \frac{(\alpha_{12}+\alpha_{22})\,(1+C_1\,\alpha_{11}+C_1\,\alpha_{21}+C_3\,\alpha_{31})}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} \\ t[3] & \frac{\alpha_{31}}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} & \frac{-C_1\,\alpha_{12}\,\alpha_{31}-C_1\,\alpha_{22}\,\alpha_{31}+\alpha_{32}+C_1\,\alpha_{11}\,\alpha_{32}+C_1\,\alpha_{21}\,\alpha_{32}}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} \end{pmatrix}$$

$$\begin{pmatrix} \omega\,(1 + C_1\,\alpha_{11} + C_1\,\alpha_{21}) & h[1] & h[2] \\ t[1] & \frac{(\alpha_{11}+\alpha_{21})\,(1+C_1\,\alpha_{11}+C_1\,\alpha_{21}+C_3\,\alpha_{31})}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} & \frac{(\alpha_{12}+\alpha_{22})\,(1+C_1\,\alpha_{11}+C_1\,\alpha_{21}+C_3\,\alpha_{31})}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} \\ t[3] & \frac{\alpha_{31}}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} & \frac{-C_1\,\alpha_{12}\,\alpha_{31}-C_1\,\alpha_{22}\,\alpha_{31}+\alpha_{32}+C_1\,\alpha_{11}\,\alpha_{32}+C_1\,\alpha_{21}\,\alpha_{32}}{1+C_1\,\alpha_{11}+C_1\,\alpha_{21}} \end{pmatrix}$$

True

The “double” meta-group

```
{β = B[ω, Sum[α10 i+j t[i] h[j], {i, 4}, {j, 4}]],
  t1 = β // dm[1, 2, 1] // dm[1, 3, 1],
  t2 = β // dm[2, 3, 2] // dm[1, 2, 1],
  t1 == t2 // Simplify
 } // βForm // ColumnForm
```

A very large output was generated. Here is a sample of it:

$$\begin{pmatrix} \omega & h[1] & h[2] & h[3] & h[4] \\ t[1] & \alpha_{11} & \alpha_{12} & \alpha_{13} & \alpha_{14} \\ t[2] & \alpha_{21} & \alpha_{22} & \alpha_{23} & \alpha_{24} \\ t[3] & \alpha_{31} & \alpha_{32} & \alpha_{33} & \alpha_{34} \\ t[4] & \alpha_{41} & \alpha_{42} & \alpha_{43} & \alpha_{44} \end{pmatrix}$$

$$\begin{pmatrix} \omega\,(1 + C_1\,\alpha_{12} + C_1\,\alpha_{13} + C_1^2\,\alpha_{12}\,\alpha_{13} + C_1\,\alpha_{23} + C_1^2\,\alpha_{12}\,\alpha_{23} + C_1^2\,\alpha_{13}\,\alpha_{32} + C_1\,C_4\,\alpha_{13}\,\alpha_{42}) & \text{h} \\ t[1] & \frac{\alpha_{11}+\alpha_{12}+2\,C_1\,\alpha_{11}\,\alpha_{12}+\ll 643\gg+\cdots}{1+C_1\,\alpha_{12}+C_1\,\alpha_{13}+C_1^2\,\alpha_{12}\,\alpha_{13}+C_1\,\alpha\cdots} \\ t[4] & \frac{\alpha_{41}+\ll\cdots}{1+\ll\cdots} \end{pmatrix}$$

$$\begin{pmatrix} \omega\,(1 + C_1\,\alpha_{12} + C_1\,\alpha_{13} + C_1^2\,\alpha_{12}\,\alpha_{13} + C_1\,\alpha_{23} + C_1^2\,\alpha_{12}\,\alpha_{23} + C_1^2\,\alpha_{13}\,\alpha_{32} + C_1\,C_4\,\alpha_{13}\,\alpha_{42}) & \text{h} \\ t[1] & \frac{\alpha_{11}+\alpha_{12}+2\,C_1\,\alpha_{11}\,\alpha_{12}+\ll 643\gg+\cdots}{1+C_1\,\alpha_{12}+C_1\,\alpha_{13}+C_1^2\,\alpha_{12}\,\alpha_{13}+C_1\,\alpha\cdots} \\ t[4] & \cdots \end{pmatrix}$$

True

Show Less | Show More | Show Full Output | Set Size Limit...

The Knot-Theoretic Equations

R2, OC, R3 and easy R4

```
{R[1, 2] Ri[3, 4],
  R[1, 2] Ri[3, 4] // dm[1, 3, 1] // dm[2, 4, 2],
  R[1, 2] Ri[3, 4] // dm[1, 3, 1] // dm[4, 2, 2]
 } // βForm
```

$$\left\{\begin{pmatrix} 1 & h[2] & h[4] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & 0 \\ t[3] & 0 & -\frac{e^{-c_3}(-1+e^{c_3})}{c_3} \end{pmatrix}, (1), (1)\right\}$$

```
{
  R[1, 2] ** Ri[1, 2],
  R[1, 3] ** R[2, 3],
  R[1, 3] ** R[2, 3] == R[2, 3] ** R[1, 3] // Simplify,
  R[3, 1] ** R[3, 2] == R[3, 2] ** R[3, 1],
  R[1, 2] ** R[1, 3] ** R[2, 3],
  R[1, 2] ** R[1, 3] ** R[2, 3] == R[2, 3] ** R[1, 3] ** R[1, 2]
 } // βForm
```

$$\left\{(1), \begin{pmatrix} 1 & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} \\ t[2] & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}, \frac{(-1+e^{c_1})(-1+e^{c_2})\,h[3]\,(-c_2\,t[1]+c_1\,t[2])}{c_1\,c_2} == 0,\right.$$

$$\left.\text{True}, \begin{pmatrix} 1 & h[2] & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} & \frac{-1+e^{c_1}}{c_1} \\ t[2] & 0 & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}, \text{True}\right\}$$

```
{
  R[3, 1] ** R[3, 2],
  R[3, 1],
  R[3, 1] // dΔ[1, 1, 2],
  R[3, 1] ** R[3, 2] == (R[3, 1] // dΔ[1, 1, 2])
 } // βForm
```

$$\left\{\begin{pmatrix} 1 & h[1] & h[2] \\ t[3] & \frac{-1+e^{c_3}}{c_3} & \frac{-1+e^{c_3}}{c_3} \end{pmatrix}, \begin{pmatrix} 1 & h[1] \\ t[3] & \frac{-1+e^{c_3}}{c_3} \end{pmatrix}, \begin{pmatrix} 1 & h[1] & h[2] \\ t[3] & \frac{-1+e^{c_3}}{c_3} & \frac{-1+e^{c_3}}{c_3} \end{pmatrix}, \text{True}\right\}$$

Hard R4

```
{
  R[1, 3] ** R[2, 3],
  R[1, 3] // dΔ[1, 1, 2],
  R[1, 3] ** R[2, 3] == (R[1, 3] // dΔ[1, 1, 2])
 } // βForm
```

$$\left\{\begin{pmatrix} 1 & h[3] \\ t[1] & \frac{-1+e^{c_1}}{c_1} \\ t[2] & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix}, \begin{pmatrix} 1 & h[3] \\ t[1] & \frac{-1+e^{c_1+c_2}}{c_1+c_2} \\ t[2] & \frac{-1+e^{c_1+c_2}}{c_1+c_2} \end{pmatrix},\right.$$

$$\left. h[3]\left(\frac{(-1+e^{c_1})\,t[1]}{c_1}+\frac{e^{c_1}(-1+e^{c_2})\,t[2]}{c_2}\right) == h[3]\left(\frac{(-1+e^{c_1+c_2})\,t[1]}{c_1+c_2}+\frac{(-1+e^{c_1+c_2})\,t[2]}{c_1+c_2}\right)\right\}$$

```
{
  V0 = B[V₀, Sum[V_{10 i+j} t[i] h[j], {i, 2}, {j, 2}]],
  V0 ** R[1, 3] ** R[2, 3],
  (R[1, 3] // dΔ[1, 1, 2]) ** V0,
  V0 ** R[1, 3] ** R[2, 3] == (R[1, 3] // dΔ[1, 1, 2]) ** V0 // Simplify
 } // βForm
```

$$\left\{\begin{pmatrix} V_0 & h[1] & h[2] \\ t[1] & V_{11} & V_{12} \\ t[2] & V_{21} & V_{22} \end{pmatrix}, \begin{pmatrix} V_0 & h[1] & h[2] & h[3] \\ t[1] & V_{11} & V_{12} & \frac{-1+e^{c_1}}{c_1} \\ t[2] & V_{21} & V_{22} & \frac{e^{c_1}(-1+e^{c_2})}{c_2} \end{pmatrix},\right.$$

$$\begin{pmatrix} V_0 & h[1] & h[2] & h[3] \\ t[1] & V_{11} & V_{12} & -\frac{(-1+e^{c_1+c_2})(-1+c_2 V_{12}-c_2 V_{21})}{c_1+c_2} \\ t[2] & V_{21} & V_{22} & \frac{(-1+e^{c_1+c_2})(1+c_1 V_{12}-c_1 V_{21})}{c_1+c_2} \end{pmatrix},$$

$$\left.\left(h[3]\left((-1+e^{c_1})\,c_2+c_1\left(-e^{c_1}(-1+e^{c_2})+(-1+e^{c_1+c_2})\,c_2\,(V_{12}-V_{21})\right)\right)(-c_2\,t[1]+c_1\,t[2])\right)\Big/\left(c_1\,c_2\,(c_1+c_2)\right) == 0\right\}$$

$$\mathbf{((-1+e^{c_1})\,c_2+c_1\,(-e^{c_1}\,(-1+e^{c_2})+(-1+e^{c_1+c_2})\,c_2\,(V_{12}-V_{21})))}$$

$$(-1+e^{c_1})\,c_2+c_1\,(-e^{c_1}\,(-1+e^{c_2})+(-1+e^{c_1+c_2})\,c_2\,(V_{12}-V_{21}))$$

$$\mathbf{Solve[(-1+e^{c_1})\,c_2+c_1\,(-e^{c_1}\,(-1+e^{c_2})+(-1+e^{c_1+c_2})\,c_2\,(V_{12})) == 0,\ V_{12}]}$$

$$\left\{\left\{V_{12}\to\frac{-e^{c_1}\,c_1+e^{c_1+c_2}\,c_1+c_2-e^{c_1}\,c_2}{(-1+e^{c_1+c_2})\,c_1\,c_2}\right\}\right\}$$